

```
clear
cd "D:\jason\workshop\Latent Class Analysis"

log using "D:\jason\workshop\Latent Class Analysis\lca3.log", replace

*****
* Model Comparison
*****

use https://www.stata-press.com/data/r16/gsem_lca1, clear

des
sum

quietly: gsem (accident play insurance stock <- ), logit lclass(C 1)
estat lcgof
estimates store oneclass

quietly: gsem (accident play insurance stock <- ), logit lclass(C 2)
estat lcgof
estimates store twoclass

quietly: gsem (accident play insurance stock <- ), logit lclass(C 3)
estat lcgof
estimates store threeclass

estimates stats oneclass twoclass threeclass

*****
* Interpretation of the best fitting model
*****

use https://www.stata-press.com/data/r16/gsem_lca1, clear

gsem (accident play insurance stock <- ), logit lclass(C 2)
estat lcprob
estat lcmean
predict classpost*, classposteriorpr

*****
* Use Margins and Marginsplot to present the LCA results
*****

use https://www.stata-press.com/data/r16/gsem_lca1, clear

quietly: gsem (accident play insurance stock <- ), logit lclass(C 2)

margins, predict(classpr class(1)) predict(classpr class(2))

marginsplot, xtitle ("") ytitle ("") ///
xlabel (1 "Class 1" 2 "Class 2" ) ///
title("Predicted Latent Class Probabilities with 95% CI")

margins, predict(classpr class(1)) predict(classpr class(2))

marginsplot, recast(bar) xtitle ("") ytitle ("") ///
xlabel (1 "Class 1" 2 "Class 2" ) ///
title("Predicted Latent Class Probabilities with 95% CI")

margins, predict(outcome(accident) class(1)) ///
predict(outcome(play) class(1)) ///
predict(outcome(insurance) class(1)) ///
predict(outcome(stock) class(1))

marginsplot, recast(bar) xtitle ("") ytitle ("") ///
xlabel (1 "accident" 2 "play" 3 "insurance" 4 "stock") ///
title("Predicted Probability of Behaviors For Class 1 with 95% CI") saving(graph1, replace)

margins, predict(outcome(accident) class(2)) ///
predict(outcome(play) class(2)) ///
```

```
predict(outcome(insurance) class(2)) ///  
predict(outcome(stock) class(2))  
  
marginsplot, recast(bar) xtitle ("") ytitle ("") ///  
xlabel (1 "accident" 2 "play" 3 "insurance" 4 "stock") ///  
title("Predicted Probability of Behaviors For Class 2 with 95% CI") saving(graph2, replace)  
  
log close
```